
1
Introduction

The hybrid controller described in the following allows a digital hostcomputer to take total control
of an Analog Paradigm Model-1 analog computer by means of a USB interface. The controller
itself is based on an AVR processor and not only allows full mode control (initial condition, operate,
halt) of the analog computer but also contains eight digitally controlled potentiometers with a 10 bit
resolution, and makes it possible to address and read out every element of the analog computer in
any mode of operation.

Figure 1.1 shows the front panel of the hybrid controller. The sixteen jacks grouped together
in the upper half are the inputs and outputs of eight digitally controller potentiometers, while the
sixteen jacks in the lower half are eight digital inputs and eight digital outputs. These digital I/O
lines are typically connected to the outputs of comparators or to the inputs of the electronic switches
of a comparator module.

The USB connector is visible on the lower far left, while the two connectors on the lower right
can be used to apply an external halt signal (typically derived from a comparator output in an
analog computer setup) and as a trigger output which can be used to trigger the x-deflection of an
oscilloscope etc. Furthermore, there are seven LEDs on the right hand side of the module which
display the current mode of operation (INITIAL, OPERATE, HALT, POTSET) as well as an OVERLOAD
condition or any other ERROR which typically results from a communication problem with the host
computer. The LED labeled USB is lit green when the USB port is connected to a host computer and
flickers during an ongoing communication.

Prof. Dr. Bernd Ulmann, 01.09.2019, Version 0.1



3

Figure 1.1: Front panel of the hybrid controller

2
Direct interaction

The simplest mode of operation is direct interaction with the hybrid controller by means of a suitable
terminal emulation such as the Serial Monitor which is part of the Arduino IDE.1 The communication
speed of the hybrid controller is set to 250000 Baud, and the interface name depends on the host
computer as well as on the particular hybrid controller. A typical device name under LINUX/Mac OS
C looks like /dev/cu.usbserial-DN050L2C.

The commands shown in table 2.1 are sent to the hybrid controller after entering them, followed
by a return-key-press if the serial monitor of the Arduino IDE is used. Please note that this line
terminator (actually a carriage return or the like, depending on the host operating system) is not
transmitted to the hybrid controller. It merely denotes the end of an input to the serial monitor.
Entering the command ?, followed by pressing the return-key yields a short help text.

2.1 A simple example

To perform a computation run of the analog computer under manual control, just enter i (followed by
pressing the return-key, as always) to set all integrators (as long as these are not externally controlled)
to their respective initial conditions. This is then followed by the command o to enter the operate
mode. This mode can then be left by either going back to initial condition or by entering h to set
the analog computer to halt mode.

Like the traditional control unit CU, the hybrid controller also allows repetitive or single operation
of the analog computer with configurable times for the initial condition and operate modes. Assume
that a given computer setup yields one solution of a simulation in 10 milliseconds and requires a further

1This can be downloaded from https://www.arduino.cc/en/Main/Software.

a Disable halt-on-overflow
A Enable halt-on-overflow
b Disable external halt
B Enable external halt
c\d{6} Set OP time for repetitive/single operation
C\d{6} Set IC time for repetitive/single operation
d[0-7] Clear digital output
D[0-7] Set digital output
e Start repetitive operation
E Start single IC/OP-cycle
F Start single IC/OP-cycle with completion message (for sync operation)
g\x{4} Set address of computing element and return its ID and value
h Halt
i Initial condition
m\x{4} Set the address of the HC module, default is 0x0090
o Operate
P\d\d{4} Set the builtin potentiometer t.o value \d{4}
q Dump digital potentiometer settings
R Read digital inputs
s Print status
S Switch to PotSet-mode
t Print elapsed OP-time
x Reset
? Print Help

Table 2.1: Commands supported by the hybrid controller (data formats are specified by regular
expressions, \d{4} denoting four decimal digits, \x{4} representing four hexadecimal digits etc.)

Prof. Dr. Bernd Ulmann, 01.09.2019, Version 0.1



6

10 ms for the initial condition phase. To get a more or less flicker free display on an oscilloscope,
the computer should be operated in repetitive mode which can be accomplished by the following
sequence of commands (comments are in red):

C000010 Set the initial condition time to 10 ms
c000010 Set the operate time to 10 ms
e Start repetitive operation

These commands are echoed by the hybrid controller with T IC=10, T OP=10, and REP-MODE
respectively (these replies are evaluated by the Perl library described in the following but are easily
readable for a human operator as well). To end the repetitive operation, either i or h can be sent to
the hybrid controller to set the analog computer to initial condition or to halt.

It is important that the times specified above are given in microseconds in a strict six-digit format!
Entering C10 would result in an error! The same holds true for other commands expecting parameters.

If the example shown above would have contained the command A prior to the start of the
repetitive operation (e), any overload condition occurring during one of the operate cycles would
cause the repetitive operation to stop immediately. In this case, the analog computer is placed into
halt mode so that the computing element being overloaded can be easily identified.

2.2 Controlling potentiometers

As already mentioned, the hybrid controller contains eight digitally controlled coe�cient potentiome-
ters with a resolution of 10 bits each. These are controlled by the P-command which expects the
number of the potentiometer to set (0 to 7) immediately followed by a four digit decimal value in the
range of 0 to 210 � 1 = 1023. To set potentiometer 0 to its mid-scale value, the command P00511
must be issued.

At power-on, the hybrid controller performs an intialization routine that sets all potentiometers
to the default value 0. It should be noted that the hybrid controller must be the last module on
the first backplane of the analog computer, so it will typically sit right next to the power supply
which occupies the rightmost position in the main chassis. This is necessary to ensure that the hybrid
controller is at its default address on the bus which is required to set the potentiometers.2

2.3 Read out operation

The hybrid controller can be used to read out the value of every computing element of the analog
computer with 16 bits of precision. All values are refered to the machine units of ±10 V, so a voltage

2If the hybrid controller is located at another slot, the m-command can be used to set a di↵erent address for it, but

this is typically not recommended for normal operation.

Module ID
PS 0
SUM8 1
INT4 2
PT8 3
CU 4
MLT8 5
MDS2 6
CMP4 7
HC 8

Table 2.2: Module types

of +5 V would be displayed as the value 0.5000. To read out the value of the first (subaddress 0)
element of the sixth (element address 5) module in the first chassis (chassis address 0) of the first
rack (rack address 0), the command g0050 would have to be issued. The first hex-nibble contains
the rack address, the second one the chassis address, the third one the slot address, and the last one
the address of the computing element on that particular module.

The output of the g-command consists of the ASCII-representation of a floating point value,
followed by a number denoting the type of the element read out. Assuming that the element on
address 0050 is a summer, having an output value of 5 V, the command above would yields the
output 0.5000 1. Table 2.2 shows all currently supported module types.

3
HyCon.pm

As useful and simple the manual operation described before is, a real hybrid computer setup requires
the digital computer to fully take control of the analog computer by means of some control program.
This is done by means of a Perl library called HyCon.pm which implements a hybrid controller class
as described in the following.

3.1 Configuration files

Every program using this library requires a configuration file in YAML-format that has the same name
as the Perl program but with extension .yml instead of .pl. A typical, yet simple configuration file
is shown in figure 3.1.

The first section contains the communications parameters of which only the name of the device
(port) should be changed as it depends on the actual setup of the host computer.

The next section, builtin dpt, specifies default values for the builtin digitally controlled poten-
tiometers. If it is missing, all eight potentiometers are initialized to 0, otherwise the eight values in
the comma separated list following values: will be used for their initialization.

The types-section maps the numeric id values returned from the various computing elements
upon readout to clear text descriptions.

The section labeled manual potentiometers lists all manual potentiometers used in a computer
setup. This is useful as the values of all these potentiometers can be read out at once using the
read mpts()-method. This functionality is typically used when some simulation required manual
changes to the coe�cient potentiometers of the analog computers. These values can then be read
out by the host computer to persist them for later analysis or the like. Please note that all elements

serial:
port: /dev/cu.usbserial-DN050L2C
bits: 8
baud: 250000
parity: none
stopbits: 1
poll_interval: 1000
poll_attempts: 200

builtin_dpt:
values: .1, .2, .3, .4, .5, .6, .7, .8

types:
0: PS
1: SUM8
2: INT4
3: PT8
4: CU
5: MLT8
6: MDS2
7: CMP4
8: HC

manual_potentiometers:
PT_8-0, PT_8-1, PT_8-2, PT_8-3: 0x0223

elements:
MUP: 0x0000
MUN: 0x0001
PT_8-0: 0x0220
PT_8-1: 0x0221
PT_8-2: 0x0222
PT_8-3: 0x0223
SUM8-0: 0x0050
SUM8-1: 0x0051
SUM8-2: 0x0052
SUM8-3: 0x0053
SUM8-4: 0x0054
SUM8-5: 0x0055
SUM8-6: 0x0056
SUM8-7: 0x0057

Figure 3.1: Example configuration file for a hybrid control program

1 use strict;
2 use warnings;
3

4 use lib ’../..’; # Path the HyCon.pm
5 use File::Basename;
6 use HyCon;
7

8 (my $config_filename = basename($0)) =~ s/\.pl$//;
9 my $ac = HyCon->new("$config_filename.yml"); # Create object

10

11 $ac->set_ic_time(500); # Set IC-time to 500 ms
12 $ac->set_op_time(1000); # Set OP-Time to 1000 ms
13 $ac->single_run(); # Perform a single computation run
14

15 # Read a value from a computing element addressed via the central bus:
16 my $element_name = ’SUM8-0’;
17 my $value = $ac->read_element($element_name);
18 print "Value = $value->{value}, Type = $value->{id}\n";

Figure 3.2: Simple test program

of the comma separated list following the manual potentiometers entry must be defined in the
elements-section of the configuration file!

The last and typically largest section is labeled elements and contains all available computing
elements (ideally only those which are actually used in some particular computer setup). The names
defined here are arbitrary and will be typically not be SUM8-0 but something like VELOCITY to enhance
readability of the control program.

3.2 A simple test program

Figure 3.2 shows a very simple test program which uses the aforementioned configuration file. Line 4
is only necessary if the HyCon.pm module is not installed in a standard location contained in @INC of
the Perl interpreter. In line 8, the name of the configuration file is derived from the current program’s
name which is then used to instantiate a HyCon-object $ac. This object, which is actually a singleton,
is used in the following to control the analog computer.

In this case, a single computation with an initial condition time of 500 ms and an operation time of
1 second is to be performed. These times are set in lines 11 and 12, followed by the invocation of the

single run()-method. After completion of this computer run, the analog computer is automatically
set to halt mode, so that results of the simulation may be read out. In this case, the value of the
element named SUM8-0 as specified in the elements-section of the corresponding configuration file
is read out in line 17. Each readout operation yields a reference to a hash which contains a value
and an id, each of which contains the actual numeric value and the element’s identification code.

4
Examples

4.1 Trajectory optimization

The following example is an extremely simple trajectory optimization problem. Here, the trajectory of
an idealized shell experiencing neither drag nor any other influences is to be parameterized by varying
its initial velocity v0 so that it hits a defined target position. The elevation angle of the cannon is
fixed.

The x- and y-components of the velocity of the shell are thus

ẋ = v0 sin(↵) and

ẏ = cos(↵)� gt.

with g and t denoting the gravitational acceleration and time. These two variables readily yield the x-
and y-components of the shell’s position by integration. The resulting setup of the analog computer
is pretty straightforward and shown in figures 4.1 and 4.2. Here, the coe�cient potentiometer labeled
DPT0 denotes the first of the eight digitally controlled potentiometers of the hybrid controller.

The time-scale factors of all three integrators are set to k0 = 103. The outputs of this circuit are
as follows:

x and y: Position of the shell – these two outputs can be used to control an oscilloscope set to
xy-mode.

�x: This is the x distance between shell and target and is used in the digital portion of the hybrid
computer program to change the initial velocity v0 of the shell in order to minimize the target
miss distance.

Figure 4.1: Setup of the analog computer for the basic trajectory problem

Figure 4.2: Analog portion of the trajectory optimization program

HLT: This logical output from a comparator is used to trigger the external halt input of the hybrid
controller when the shell hits ground level. Therefore it is necessary that the height of the
cannon satisfies y0 > 0. Otherwise the comparator would trigger halt before the actual flight
of the shell begins.

The digital portion of the hybrid computer setup consists of the YAML configuration file shown
in figure 4.3 and its accompanying Perl program shown in figure 4.4. The sections serial and
builtin dpt are standard. The manual potentiometers- and elements-sections only list those
computing elements which are used in this setup. Note that all elements have been given symbolic
names like delta x instead of S0120, SUM8-0 etc.

The digital computer program as shown in figure 4.4 is pretty straight-forward: Since the HyCon.pm
module was not installed in a standard location for this demonstration, the Perl interpreter has to
be notified to include an additional directory into its @INC-variable (line 4). This program displays a
status line in the first line of the screen which is getting updated during the simulation run. Therefore,
the Term::ANSIScreen package is used which implements (among many other useful things) a clear
screen routine. Setting the special variable $| to 1 disables output bu↵ering of stdout which allows
us to update this status line periodically. All of this is done in lines 6 and 9–11.

The analog computer is setup in lines 16–19. Changing the address of the hybrid controller to
hexadecimal 0080 is typically not necessary and results from the fact that this example has been
implemented on a very early prototype of the Analog Paradigm Model-1 analog computer which has
some quirks regarding the addressing scheme. In line 17 the external halt input of the hybrid controller
is enabled. The idea is that a simulation runs as long as the shell requires to hit “ground” which is
detected by a comparator which is connected to the HALT input of the controller.

Since k0 = 103, an initial condition time of 1 ms as set in line 18 is more than su�cient. The
operating time of 1000 ms is unrealistically high but such a large value won’t do any harm in this
context since one simulation run ends as soon as the comparator detects the shell hitting ground
level. In fact, any value for this time frame (line 19) that exceeds the maximum run time of a single
run in the worst case would be su�cient.

The idea behind the parameter variation loop spanning lines 22–35 is extremely simple: If the
shell’s flight path is too short, v0 is increased, if it is too far, v0 is decreased accordingly. This change
in v0 is made a bit dynamical by taking the hit miss distance into account. If the distance to the
target is less than the value of $epsilon, the loop is terminated, and the current potentiometer
settings are dumped to the screen.

Figure 4.5 shows a typical output of the digital portion of the hybrid computer setup. It took 41
one trials to determine a suitable v0 which allows the shell to miss the target by only 0.0008 (arbitrary
units). Following this status line the values of all relevant manual potentiometers are shown.

Figure 4.6 shows a long-term exposure photograph of the output on the attached oscilloscope
screen. The varying step size by which v0 is changed is clearly visible. Please note that this hybrid
computer setup is neither too realistic nor too astute, it’s only purpose is to serve as an introductory

1 serial:
2 port: /dev/cu.usbserial-DN050L2C
3 bits: 8
4 baud: 250000
5 parity: none
6 stopbits: 1
7 poll_interval: 10
8 poll_attempts: 20000
9 builtin_dpt:

10 values: 0, 0, 0, 0, 0, 0, 0, 0
11 types:
12 0: PS
13 1: SUM8
14 2: INT4
15 3: PT8
16 4: CU
17 5: MLT8
18 6: MDS2
19 7: CMP4
20 8: HC
21 manual_potentiometers: cos_alpha, sin_alpha, PT_y0, PT_x_scale, PT_x_target, g
22 elements:
23 cos_alpha: 0x0030
24 sin_alpha: 0x0031
25 PT_y0: 0x0032
26 PT_x_scale: 0x0033
27 PT_x_target: 0x0034
28 g: 0x0035
29

30 delta_x: 0x0120
31 minus_y: 0x0121
32

33 x: 0x0160
34 int_g: 0x0161
35 y: 0x0162

Figure 4.3: Configuration file for the simple trajectory optimization

1 use strict;
2 use warnings;
3

4 use lib ’../..’; # Path the HyCon.pm
5 use File::Basename;
6 use Term::ANSIScreen;
7 use HyCon;
8

9 $| = 1;
10 my $console = Term::ANSIScreen->new();
11 $console->Cls();
12

13 (my $config_filename = basename($0)) =~ s/\.pl$//;
14 my $ac = HyCon->new("$config_filename.yml"); # Create object
15

16 $ac->set_address(’0080’); # Just for the prototype system
17 $ac->enable_ext_halt(); # This is essential
18 $ac->set_ic_time(1);
19 $ac->set_op_time(1000); # This is a limit that will never be reached
20

21 my ($counter, $v0, $delta_v, $epsilon) = (1, 0, .1, .001);
22 while (1) {
23 $ac->set_pt(0, $v0);
24

25 my $halt = $ac->single_run_sync();
26 $halt = 0 unless defined($halt);
27 my $delta_x = $ac->read_element(’delta_x’)->{value};
28

29 printf("H: $halt\tTrial: %06d\tv0 = %+0.4f\tDelta x = %+0.4f\r",
30 $counter++, $v0, $delta_x);
31

32 last if abs($delta_x) < $epsilon;
33

34 $v0 += $delta_x * $delta_v;
35 }
36 print "\n\n";
37 my $pot_settings = $ac->read_mpts();
38 print "$_:\t$pot_settings->{$_}{value}\n" for sort(keys(%$pot_settings));

Figure 4.4: Digital portion of the simple trajectory optimization

1 H: 1 Trial: 000034 v0 = +0.4204 Delta x = +0.0007
2

3 PT_x_scale: 0.2186
4 PT_x_target: -0.2191
5 PT_y0: -0.5250
6 cos_alpha: -0.2975
7 g: -0.1074
8 sin_alpha: -0.2322

Figure 4.5: Typical output of a simulation run on the digital computer

Figure 4.6: Screenshot from the running trajectory optimization program

example to hybrid computer programming. A more sophisticated implementation might take the
velocity dependent drag into account.1 Furthermore, it is not realistic to change the muzzle velocity
v0 of a shell as the only parameter of a simulation. Changing the elevation angle of the cannon, and
thus changing the values of cos(↵) and sin(↵) would be more realistic. This is left as an exercise to
the reader.2 :-)

1See http://analogparadigm.com/downloads/alpaca_9.pdf for an example of such a computation.
2Or to a later application note. . .

A
HyCon.pm

HyCon.pm

1 #
2 # The HyCon-package provides an object oriented interface to the HYCON
3 # hybrid controller for the Analogparadigm Model-1 analog computer.
4 #
5 # 06-AUG-2016 B. Ulmann Initial version
6 # 07-AUG-2016 B. Ulmann Added extensive error checking, changed
7 # c-/C-commands for easier interfacing
8 # 08-AUG-2016 B. Ulmann Analog calibration capability added
9 # 31-AUG-2016 B. Ulmann Support of digital potentiometers

10 # 01-SEP-2016 B. Ulmann Initial potentiometer setting based on
11 # configuration file etc.
12 # 13-MAY-2017 B. Ulmann Start adaptation to new, AVR2560-based hybrid
13 # controller with lots of new features
14 # 16-MAY-2017 B. Ulmann single_run_sync() implemented
15 # 08-FEB-2018 B. Ulmann Changed read_element to expect the name of a
16 # computing element instead of its address
17 # 01-SEP-2018 B. Ulmann Adapted to the final implementation of the
18 # hybrid controller (version 0.4)
19 # 02-SEP-2018 B. Ulmann Bug fixes, get_response wasn’t implemented too
20 # cleverly, it is now much faster than before :-)
21 #
22

23 package HyCon;
24

25 =pod
26

27 =head1 NAME
28

29 HyCon - Hybrid controller for analog computers
30

31 =head1 VERSION
32

33 This document refers to version 0.4 of HyCon
34

35 =head1 SYNOPSIS
36

37 use strict;
38 use warnings;
39

40 use File::Basename;
41 use HyCon;
42

43 (my $config_filename = basename($0)) =~ s/\.pl$//;
44 print "Create object...\n";
45 my $ac = HyCon->new("$config_filename.yml");
46

47 $ac->set_ic_time(500); # Set IC-time to 500 ms
48 $ac->set_op_time(1000); # Set OP-Time to 1000 ms
49 $ac->single_run(); # Perform a single computation run
50

51 # Read a value from a computing element addressed via the central bus:
52 my $element_name = ’SUM8-0’;
53 my $value = $ac->read_element($element_name);
54

55 =head1 DESCRIPTION
56

57 This module implements a simple interface to the a hybrid controller which
58 interfaces an analog computer to a digital computer and thus allows true
59 hybrid computation.
60

61 =cut

Prof. Dr. Bernd Ulmann, 01.09.2019, Version 0.1



20

62

63 use strict;
64 use warnings;
65

66 our $version = 0.4;
67

68 use YAML qw(LoadFile);
69 use Carp qw(confess cluck);
70 use Device::SerialPort;
71 use Time::HiRes qw(usleep);
72

73 use Data::Dumper;
74

75 use constant {
76 DIGITAL_OUTPUT_PORTS => 8,
77 DIGITAL_INPUT_PORTS => 8,
78 BUILTIN_DPT => 8,
79 BUILTIN_DPT_RESOLUTION => 10,
80 };
81

82 my $instance;
83

84 =head1 Functions and methods
85

86 =head2 new($filename)
87

88 This function generates a HyCon-object. Currently there is only one hybrid
89 controller supported, so this is, in fact, a singleton and every subsequent
90 invocation will cause a fatal error. This function expects a path to a YAML
91 configuration file of the following structure:
92

93 config.yml:
94 serial:
95 port: /dev/tty.usbmodem621
96 bits: 8
97 baud: 250000
98 parity: none
99 stopbits: 1

100 poll_interval: 1000

101 poll_attempts: 200
102 builtin_dpt:
103 values: .1, .2, .3, .4, .5, .6, .7, .8
104 types:
105 0: PS
106 1: SUM8
107 2: INT4
108 3: PT8
109 4: CU
110 5: MLT8
111 6: MDS2
112 7: CMP4
113 8: HC
114 elements:
115 Y_DDOT: 0x0100
116 Y_DOT: 0x0101
117 PT_8-0: 0x0220
118 PT_8-1: 0x0221
119 PT_8-2: 0x0222
120 PT_8-3: 0x0223
121 PT_8-4: 0x0224
122 PT_8-5: 0x0225
123 PT_8-6: 0x0226
124 PT_8-7: 0x0227
125 manual_potentiometers:
126 PT_8-0, PT_8-1, PT_8-2, PT_8-3, PT_8-4, PT_8-5, PT_8-6, PT_8-7
127

128 The setup shown above will not fit your particular configuration, it just
129 serves as an example. The remaining parameters nevertheless apply in
130 general and are mostly self-explanatory. poll_interval and poll_attempts
131 control how often this interface will poll the hybrid controller to get a
132 response to a command. The values shown above are overly pessimistic but
133 won’t hurt during normal operation.
134

135 The section labeled ’builtin_dpt’ contains data to setup and control the
136 digital potentiometers of the hybrid controller. The initial values of
137 these potentiometers can be specified by a list containing eight entries
138 following ’values’. If this part is missing, initial values of 0 are
139 assumed. The new() function will set all digitally controlled

Prof. Dr. Bernd Ulmann, 01.09.2019, Version 0.1



22

140 potentiometers of the hybrid computer according to these data upon
141 invocation.
142

143 If the number of values specified in the array ’values’ does not match the
144 number of configured potentiometers, the function will abort.
145

146 The types-section contains the mapping of the devices types as returned by
147 the analog computer’s readout system to their module names.
148

149 The elements-section contains a list of computing elements defined by an
150 arbitrary name and their respective address in the computer system. Calling
151 read_all_elements() will switch the computer into HALT-mode, read the
152 values of all elements in this list and return a reference to a hash
153 containing all values and IDs of the elements read.
154

155 Ideally, all manual potentiometers are listed following
156 manual_potentiometers which is used for automatic readout of the settings
157 of these potentiometers by calling read_mpts(). This is useful if a
158 simulation has been parameterized manually and these parameters are then
159 required for documentation purposes or the like. Caution: All
160 potentiometers to be read out by read_mpts() must be defined in the
161 elements-section!
162

163 The new() function will clear the communication buffer of the hybrid
164 controller by reading and discarding and data until a timeout will be
165 reached. This is currently as long as the product of poll_interval and
166 poll_attempts.
167

168 =cut
169

170 sub new {
171 my ($class, $config_filename) = @_;
172

173 confess "Only one instance of a HyCon-object at a time is supported!"
174 if $instance++;
175

176 my $config = LoadFile($config_filename) or
177 confess "Could not read configuration YAML-file: $!";
178

179 my $port = Device::SerialPort->new($config->{serial}{port}) or
180 confess "Unable to open USB-port: $!\n";
181 $port->databits($config->{serial}{bits});
182 $port->baudrate($config->{serial}{baud});
183 $port->parity($config->{serial}{parity});
184 $port->stopbits($config->{serial}{stopbits});
185

186 # If no poll-interval is specified, use 1000 microseconds
187 $config->{serial}{poll_interval} //= 1000;
188 $config->{serial}{poll_attempts} //= 200; # and 200 such intervals.
189

190 # Get rid of any data which might still be in the serial line buffer
191 my $attempt;
192 $port->write(’x’); # Reset the hybrid controller
193 while (++$attempt < $config->{serial}{poll_attempts}) {
194 my $data = $port->lookfor();
195 last if $data eq ’RESET’;
196 usleep($config->{serial}{poll_interval});
197 }
198

199 # Create array of initial potentiometer values - if there is nothing
200 # specified assume zero:
201 my $pv = defined($config->{builtin_dpt}{values})
202 ? [ split(/\s*,\s*/, $config->{builtin_dpt}{values}) ]
203 : [ map{0}(1 .. BUILTIN_DPT) ];
204

205 # Create the actual object
206 my $object = bless(my $self = {
207 port => $port,
208 poll_interval => $config->{serial}{poll_interval},
209 poll_attempts => $config->{serial}{poll_attempts},
210 builtin_dpt => {
211 number => BUILTIN_DPT,
212 resolution => BUILTIN_DPT_RESOLUTION,
213 values => $pv,
214 },
215 elements => $config->{elements},
216 types => $config->{types},
217 times => {

218 ic_time => -1,
219 op_time => -1,
220 },
221 manual_potentiometers =>
222 [ split(/\s*,\s*/, $config->{manual_potentiometers}) ],
223 }, $class);
224

225 # Initial potentiometer setup
226 set_pt($object, $_, $pv->[$_]) for (0 .. BUILTIN_DPT - 1);
227

228 return $object;
229 }
230

231 =head2 get_response()
232

233 In some cases, e.g. external HALT conditions, it is necessary to query the
234 hybrid controller for any messages which may have occured since the last
235 command. This can be done with this method - it will poll the controller
236 for a period of poll_interval times poll_attemps microseconds. If this
237 time-out value is not suitable, a different value in milliseconds can be
238 supplied as first argument of this method. If this argument is zero or
239 negative, get_response will wait indefinitely for a response from the
240 hybrid controller.
241

242 =cut
243

244 sub get_response {
245 my ($self, $timeout) = @_;
246 $timeout = $self->{poll_interval} unless defined($timeout);
247

248 my $attempt;
249 do {
250 my $response = $self->{port}->lookfor();
251 return $response if $response;
252 # If we poll indefinitely, there is no need to wait at all
253 usleep($timeout) if $timeout > 0;
254 } while ($timeout < 1 or ++$attempt < $self->{poll_attempts});
255 }
256

257 =head2 ic()
258

259 This method switches the analog computer to IC (initial condition) mode
260 during which the integrators are (re)set to their respective initial value.
261 Since this involves charging a capacitor to a given value, this mode should
262 be activated for the right duration as required by the analog computer being
263 controlled. Especially on historic machines it is not uncommon to have
264 IC-durations of up to a second when a given computer setup includes
265 integrators set to time constant 1. Refer to the analog computer’s
266 documentation for detailed information on setup times.
267

268 ic() and the two following methods should not be used normally when timing
269 is critical. Instead, IC- and OP-times should be setup explicitly (see
270 below) and then a single-run should be initiated which will be under
271 control of the hybrid controller which takes care for sub-millisecond
272 precision with respect to timing issues.
273

274 =head2 op()
275

276 This method switches the analog computer to OPerating-mode.
277

278 =head2 halt()
279

280 Calling this method causes the analog computer to switch to HALT-mode. In
281 this mode the integrators are halted and store their last value. After
282 calling halt() it is possible to return to OP-mode by calling op() again.
283 Depending on the analog computer being controlled, there will be a more or
284 less substantial drift of the integrators in HALT-mode, so it is advisable
285 to keep the HALT-periods as short as possible to minimize errors.
286

287 A typical operation cycle may look like this: IC-OP-HALT-OP-HALT-OP-HALT.
288 This would start a single computation with the possibility of reading
289 values from the analog computer during the HALT-intervals.
290

291 Another typical cycle is called "repetitive operation" and looks like this:
292 IC-OP-IC-OP-IC-OP... This is normally used with the integrators set to fast
293 time-constants and allows to display a solution as a more or less flicker
294 free curve on an oscilloscope for example.
295

296 =head2 enable_ovl_halt()
297

298 During a normal computation on an analog computation there should be no
299 overloads of summers or integrators. Such overload conditions are either a
300 sign of a machine failure or (more common) an erroneous computer setup
301 (normally caused by wrong scaling of the underlying equations). To catch
302 such problems it is usually a good idea to switch the analog computer
303 automatically to HALT-mode when an overload occurs. This is done by calling
304 this method during the setup phase.
305

306 =head2 disable_ovl_halt()
307

308 Calling this method will disable the automatic halt-on-overload
309 functionality of the hybrid controller.
310

311 =head2 enable_ext_halt()
312

313 Sometimes it is necessary to halt a computation when some condition is
314 fulfilled (some value reached etc.). This is normally detected by a
315 comparator used in the analog computer setup. The hybrid controller
316 features an EXT-HALT input jack that can be connected to such a comparator.
317 After calling this method, the hybrid controller will switch the analog
318 computer from OP-mode to HALT as soon as the input signal patched to this
319 input jack goes high.
320

321 =head2 disable_ext_halt()
322

323 This method disables the HALT-on-overflow feature of the hybrid controller.
324

325 =head2 single_run()
326

327 Calling this method will initiate a so-called "single-run" on the analog
328 computer which automatically performs the sequence IC-OP-HALT. The times
329 spent in IC- and OP-mode are specified with the methods set_ic_time() and
330 set_op_time() (see below).
331

332 It should be noted that the hybrid controller will not be blocked during
333 such a single-run - it is still possible to issue other commands to read or
334 set ports etc.

335

336 =head2 single_run_sync()
337

338 This function behaves quite like single_run() but waits for the termination
339 of the single run, thus blocking any further program execution. This method
340 returns true if the single-run mode was terminated by an external halt
341 condition. Otherwise undef is returned.
342

343 =head2 repetitive_run()
344

345 This initiates repetitive operation, i.e. the analog computer is commanded
346 to perform an IC-OP-IC-OP-... sequence. The hybrid controller will also not
347 block during this run. To terminate a repetitive run, either ic() or halt()
348 may be called, depending on the mode the analog computer should stop. Note
349 that these methods act immediately and will interrupt any ongoing IC- or
350 OP-period of the analog computer.
351

352 =head2 pot_set()
353

354 This function switches the analog computer to POTSET-mode i.e. the
355 integrators are set implicitly to HALT, while all (manual) potentiometers
356 are connected to +1 on their respective input side. This mode can be used
357 to readout the current settings of the potentiometers.
358

359 =cut
360

361 # Create basic methods
362 my %methods = (
363 ic => [’i’, ’^IC’], # Switch AC to IC-mode
364 op => [’o’, ’^OP’], # Switch AC to OP-mode
365 halt => [’h’, ’^HALT’], # Switch AC to HALT-mode
366 disable_ovl_halt => [’a’, ’^OVLH=DISABLED’], # Disable HALT-on-overflow
367 enable_ovl_halt => [’A’, ’^OVLH=ENABLED’], # Enable HALT-on-overflow
368 disable_ext_halt => [’b’, ’^EXTH=DISABLED’], # Disable external HALT
369 enable_ext_halt => [’B’, ’^EXTH=ENABLED’], # Enable external HALT
370 repetitive_run => [’e’, ’^REP-MODE’], # Switch to RepOp
371 single_run => [’E’, ’^SINGLE-RUN’], # One IC-OP-HALT-cycle
372 pot_set => [’S’, ’^PS’], # Activate POTSET-mode
373 );

374

375 eval (’
376 sub ’ . $_ . ’ {
377 my ($self) = @_;
378 $self->{port}->write("’ . $methods{$_}[0] . ’");
379 my $response = get_response($self);
380 confess "No response from hybrid controller! Command was \’’ .
381 $methods{$_}[0] . ’\’." unless $response;
382 confess "Unexpected response from hybrid controller:\\n\\tCOMMAND=\’’ .
383 $methods{$_}[0] . ’\’, RESPONSE=\’$response\’, PATTERN=\’’ .
384 $methods{$_}[1] . ’\’\\n"
385 if $response !~ /’ . $methods{$_}[1] . ’/;
386 }
387 ’) for keys(%methods);
388

389 sub single_run_sync() {
390 my ($self) = @_;
391 $self->{port}->write(’F’);
392 my $response = get_response($self);
393 confess "No Response from hybrid controller! Command was ’F’"
394 unless $response;
395 confess "Unexpected response:\n\tCOMMAND=’F’, RESPONSE=’$response’\n"
396 if $response !~ /^SINGLE-RUN/;
397 my $timeout = 1.1 * ($self->{times}{ic_time} + $self->{times}{op_time});
398 $response = get_response($self);
399 confess "No Response during single_run_sync within $timeout ms"
400 unless $response;
401 confess "Unexpected response after single_run_sync: ’$response’\n"
402 if $response !~ /^EOSR/ and $response !~ /^EOSRHLT/;
403 # Return true if the run was terminated by an external halt condition
404 return $response =~ /^EOSRHLT/;
405 }
406

407 =head2 set_ic_time($milliseconds)
408

409 It is normally advisable to let the hybrid controller take care of timing the
410 analog computer modes of operation as the communication with the digital host
411 introduces quite some jitter. This method sets the time the analog computer
412 will spend in IC-mode during a single- or repetitive run. The time is

413 specified in milliseconds and must be positive and can not exceed 999999
414 milliseconds due to limitations of the hybrid controller itself.
415

416 =cut
417

418 # Set IC-time
419 sub set_ic_time {
420 my ($self, $ic_time) = @_;
421 confess ’IC-time out of range - must be >= 0 and <= 999999!’
422 if $ic_time < 0 or $ic_time > 999999;
423 my $pattern = "^T_IC=$ic_time\$";
424 my $command = sprintf("C%06d", $ic_time);
425 $self->{port}->write($command);
426 my $response = get_response($self);
427 confess ’No response from hybrid controller!’ unless $response;
428 confess "Unexpected response: ’$response’, expected: ’$pattern’"
429 if $response !~ /$pattern/;
430 $self->{times}{ic_time} = $ic_time;
431 }
432

433 =head2 set_op_time($milliseconds)
434

435 This method specifies the duration of the OP-cycle(s) during a single- or
436 repetitive analog computer run. The same limitations hold with respect to the
437 time specified as for the set_ic_time() method.
438

439 =cut
440

441 # Set OP-time
442 sub set_op_time {
443 my ($self, $op_time) = @_;
444 confess ’OP-time out of range - must be >= 0 and <= 999999!’
445 if $op_time < 0 or $op_time > 999999;
446 my $pattern = "^T_OP=$op_time\$";
447 my $command = sprintf("c%06d", $op_time);
448 $self->{port}->write($command);
449 my $response = get_response($self);
450 confess ’No response from hybrid controller!’ unless $response;
451 confess "Unexpected response: ’$response’, expected: ’$pattern’"

452 if $response !~ /$pattern/;
453 $self->{times}{op_time} = $op_time;
454 }
455

456 =head2 read_element($name)
457

458 This function expects the name of a computing element specified in the
459 configuation YML-file and applies the corresponding 16 bit value $address to
460 the address lines of the analog computer’s bus system, asserts the active-low
461 /READ-line, reads one value from the READOUT-line, and de-asserts /READ again.
462 read_element(...) returns a reference to a hash with the keys ’value’ and
463 ’id’.
464

465 =cut
466

467 sub read_element {
468 my ($self, $name) = @_;
469 my $address = hex($self->{elements}{$name});
470 confess "Computing element $name not configured!\n"
471 unless defined($address);
472 $self->{port}->write(’g’ . sprintf("%04X", $address & 0xffff));
473 my $response = get_response($self);
474 confess ’No response from hybrid controller!’ unless $response;
475 my ($value, $id) = split(/\s+/, $response);
476 $id = $self->{types}{$id & 0xf} || ’UNKNOWN’;
477 return { value => $value, id => $id};
478 }
479

480 =head2 read_all_elements()
481

482 The routine read_all_elements() switches the computer to HALT and reads the
483 current values from all elements listed in the elements-section of the
484 configuration file. It returns a reference to a hash containing all elements
485 with their associated values and IDs.
486

487 =cut
488

489 sub read_all_elements {
490 my ($self) = @_;

Prof. Dr. Bernd Ulmann, 01.09.2019, Version 0.1



31

491 $self->halt();
492 my %result;
493 for my $key (sort(keys(%{$self->{elements}})))
494 {
495 my $result = $self->read_element($key);
496 $result{$key} = { value => $result->{value}, id => $result->{id} };
497 }
498 return \%result;
499 }
500

501 =head2 read_digital()
502

503 In addition to the analog input channels mentioned above, the hybrid
504 controller also features digital inputs (two) which can be used to read out
505 the state of comparators or other logic elements of the analog computer being
506 controlled. This method also returns an array-reference containing values of
507 0 or 1.
508

509 =cut
510

511 # Read digital inputs
512 sub read_digital {
513 my ($self) = @_;
514 $self->{port}->write(’R’);
515 my $response = get_response($self);
516 confess ’No response from hybrid controller!’ unless $response;
517 my $pattern = ’^’ . ’\d+\s+’ x (DIGITAL_INPUT_PORTS - 1) . ’\d+’;
518 confess "Unexpected response: ’$response’, expected: ’$pattern’"
519 if $response !~ /$pattern/;
520 return [ split(/\s+/, $response) ];
521 }
522

523 =head2 digital_output($port, $value)
524

525 The hybrid controller also features digital outputs (two) which can be used to
526 control electronic/relay switches in the analog computer being controlled.
527 Calling digital_output(0, 1) will set the first (0) digital output to 1 etc.
528

529 =cut

530

531 # Set/reset digital outputs
532 sub digital_output {
533 my ($self, $port, $state) = @_;
534 confess ’$port must be >= 0 and < ’ . DIGITAL_OUTPUT_PORTS
535 if $port < 0 or $port > DIGITAL_OUTPUT_PORTS;
536 $self->{port}->write(($state ? ’D’ : ’d’) . $port);
537 }
538

539 =head2 read_mpts()
540

541 Calling read_mpts() returns a reference to a hash containing the current
542 settings of all manual potentiometers listed in the
543 manual_potentiometers-section in the configuration file. To accomplish this,
544 the analog computer is switched to POTSET-mode (implying HALT for the
545 integrators). In this mode, all inputs of potentiometers (apart from "free"
546 potentiometers unless their second input is patched to AGND) are connected to
547 +1, so that their current setting can be read out.
548

549 =cut
550

551 sub read_mpts {
552 my ($self) = @_;
553 $self->pot_set();
554 my %result;
555 for my $key (@{$self->{manual_potentiometers}}) {
556 my $result = $self->read_element($key);
557 $result{$key} = { value => $result->{value}, id => $result->{id} };
558 }
559 return \%result;
560 }
561

562 =head2 set_pt($address, $value)
563

564 To set a digital potentiometer, set_pt() is called. The first argument is the
565 address of the potentiometer to be set (0 <= number < number-of-potentiometers
566 as specified in the potentiometers section in the configuration YML-file), the
567 second argument is a floatingpoint value 0 <= v <= 1. If either the address or
568 the value is out of bounds, the function will die.

569

570 =cut
571

572 sub set_pt {
573 my ($self, $address, $value) = @_;
574 confess ’Addr must be >= 0 and < ’ . $self->{builtin_dpt}{number} .
575 " it is $address"
576 if $address < 0 or $address >= $self->{builtin_dpt}{number};
577 confess ’$value must be >= 0 and <= 1’ if $value < 0 or $value > 1;
578

579 # Convert value to an integer suitable to setting the potentiometer and
580 # generate fixed length strings for the parameters address and value:
581 $value = sprintf(’%04d’,
582 int($value * (2 ** $self->{builtin_dpt}{resolution} - 1)));
583 $address = sprintf(’%d’, $address);
584 $self->{port}->write("P$address$value"); # Send command
585 my $response = get_response($self); # Get response
586 confess ’No response from hybrid controller!’ unless $response;
587 my ($raddress, $rvalue) = $response =~ /^P(\d+)=(\d+)$/;
588 confess "set_pt failed! $address vs. $raddress, $value vs. $rvalue"
589 if $address != $raddress or $value != $rvalue;
590 }
591

592 =head2 get_status()
593

594 Calling get_status() yields a reference to a hash containing all current
595 status information of the hybrid controller. A typical hash structure
596 returned may look like this:
597

598 $VAR1 = {
599 ’IC-time’ => ’500’,
600 ’MODE’ => ’HALT’,
601 ’OP-time’ => ’1000’,
602 ’STATE’ => ’NORM’,
603 ’OVLH’ => ’DIS’,
604 ’EXTH’ => ’DIS’
605 };
606

607 In this case the IC-time has been set to 500 ms while the OP-time is set to

608 one second. The analog computer is currently in HALT-mode, and the hybrid
609 controller is in its normal state, i.e. it is not currently performing a
610 single- or repetitive-run. HALT on overload and external HALT are both
611 disabled.
612

613 =cut
614

615 # Get status returns a hash-reference
616 sub get_status {
617 my ($self) = @_;
618 $self->{port}->write(’s’);
619 my $response = get_response($self);
620 confess ’No response from hybrid controller!’ unless $response;
621 my %state;
622 for my $entry (split(/\s*,\s*/, $response)) {
623 my ($parameter, $value) = split(/\s*=\s*/, $entry);
624 $state{$parameter} = $value;
625 }
626 return \%state;
627 }
628

629 =head2 get_op_time()
630

631 In some applications it is useful to be able to determine how long the analog
632 computer has been in OP-mode. As time as such is the only so-called free
633 variable of integration in an analog-electronic analog computer, it is a
634 central parameter to know. Imagine that some integration is being performed by
635 the analog computer and the time which it took to reach some threshold value
636 is being investigated. In this case, the hybrid controller would be configured
637 so that external-HALT is enabled. Then the analog computer would be placed to
638 IC-mode and then to OP-mode. After an external HALT has been triggered by some
639 comparator of the analog commputer, the hybrid controller will switch the
640 analog computer to HALT-mode immediately. Afterwards, the time the analog
641 computer spent in OP-mode can be determined by calling this method. The time
642 will be returned in microseconds (the resolution should be +/- 3 to 4
643 microseconds).
644

645 =cut
646

647 # Get current time the AC spent in OP-mode
648 sub get_op_time {
649 my ($self) = @_;
650 $self->{port}->write(’t’);
651 my $response = get_response($self);
652 confess ’No response from hybrid controller!’ unless $response;
653 my $pattern = ’t_OP=\-?\d*’;
654 confess "Unexpected response: ’$response’, expected: ’$pattern’"
655 if $response !~ /$pattern/;
656 my ($time) = $response =~ /=\s*(\-?\d+)$/;
657 return $time ? $time : -1;
658 }
659

660 =head2 reset()
661

662 The reset() method resets the hybrid controller to its initial setup. This
663 will also reset all digital potentiometer settings including their number!
664 During normal operations it should not be necessary to call this method which
665 was included primarily to aid debugging.
666

667 =cut
668

669 sub reset {
670 my ($self) = @_;
671 $self->{port}->write(’x’);
672 my $response = get_response($self);
673 confess ’No response from hybrid controller!’ unless $response;
674 confess "Unexpected response: ’$response’, expected: ’RESET’"
675 if $response ne ’RESET’;
676 }
677

678 =head2 set_address(address)
679

680 set_address() is used to set the hybrid controller to a different address than
681 its default address of 0x0090. The hybrid controller requires its own address
682 on the backplane in order to set the builtin digital potentiometers. If the
683 controller is placed into another slot than the last one of the main backplane
684 (which is not recommended), then this method has to be called before any
685 changes to the builtin digitally controlled potentiometers are made. Caution:

686 In this case, setting these potentiometers to their default values as
687 specified in the corresponding configuration YML-file will not succeed! The
688 address has to be specified in hexadecimal notation with four digits (padded
689 on the left with zeros if necessary).
690

691 =cut
692

693 sub set_address() {
694 my ($self, $address) = @_;
695 $self->{port}->write("m$address");
696 my $response = get_response($self);
697 confess ’No response from hybrid controller!’ unless $response;
698 my ($value) = $response =~ /^MY_ADDR=(.+)$/;
699 confess "Unexpected response: ’$response’, expected: ’MY_ADDR=...’"
700 unless defined($value);
701 $_ =~ s/^0+// for $address, $value;
702 confess "Address returned ($value) differs from address sent ($address)!"
703 unless $address == $value;
704 }
705

706 =head1 Examples
707

708 The following example initates a repetitive run of the analog computer with 20
709 ms of operating time and 10 ms IC time:
710

711 use strict;
712 use warnings;
713

714 use File::Basename;
715 use HyCon;
716

717 (my $config_filename = basename($0)) =~ s/\.pl$//;
718 my $ac = HyCon->new("$config_filename.yml");
719

720 $ac->set_op_time(20);
721 $ac->set_ic_time(10);
722

723 $ac->repetitive_run();
724 =cut

725

726 =head1 AUTHOR
727

728 Dr. Bernd Ulmann <lt>ulmann@analogparadigm.com<gt>
729

730 =cut
731

732 return 1;
HyCon.pm

